21. Строковый тип

Классификация: нестандартный, структурированный (сложный) тип.

Имя определяет программист.

3. Структурная организация

Данные строкового типа представляют собой строки символов. Строковый тип - массив, элементами которого являются данные литерного типа CHAR. Элементы этого массива пронумерованы, начиная с 0. Сами символы располагаются, начиная с элемента с индексом 1. В нулевом элементе указывается динамическая длина строки (количество символов, которые в данный момент времени хранятся в массиве). Так как все элементы массива одинаковы по размеру, то под динамическую длину, как и под символ, отводится 1 байт. Поэтому строка символов не может иметь длину большую 255. Строка длиной 0 - пустая строка.

Рис. 21.1. – Структура строки символов

Требование однотипности элементов массива приводит к тому, что в элементе с индексом 0 располагается символ, код которого равняется количеству символов, существующих в строке в данный момент времени.

0. Определение типа

Рис. 21.2. – Определение строкового типа

В определении типа (рис. 21.2), если нужно зарезервировать строку длиной менее 255 символов, в квадратных скобках указывается необходимая длина.

Примеры определения строковых типов и переменных этих типов:

TYPE  TS10=STRING[10];


TS=STRING;

VAR A:TS10;

    B:TS;

В структурном программировании на языке Паскаль при использовании данных строкового типа этот тип не определяется в разделе нестандартных типов, а при описании переменных используется непосредственное определение строкового типа. Поэтому предыдущий пример в тексте программы должен выглядеть так

VAR A: STRING[10];

    B: STRING;

1. Множество значений

Включает все значения, которые могут быть в структуре заданной в определении типа.

2. Множество операций

Со строкой символов можно работать, как с одномерным массивом из символов в пределах заполненных элементов этого массива. Т.е. индексы находятся в диапазоне от 0 по значение ORD(<имя переменной типа строка>[0]).

Например, имеется VAR S:STRING и выполнено действие S:='реформа'. После выполнения оператора присваивания  индексы  массива S находятся в пределах  0..7 .Для того, чтобы сформировать в переменной значение 'информатика', необходимо выполнить следующие действия:

          S[0]:=CHR(11);{формирование динамической длины строки}

          S[1]:='и';

          S[2]:='н';

          S[8]:='т';

          S[9]:='и';

          S[10]:='к';

          S[11]:='а' .

Однако, для строкового типа определены и свои операции.

2.1. Операция конкатенации

Это бинарная операция со знаком +. Здесь операнды и результат любого строкового типа. В результате выполнения операции к строке, являющейся первым операндом, приклеивается вторая строка. Если длина полученной строки превышает 255, то она усекается справа до 255.

Например, в результате выполнения операции 'инфор'+'матика' получается значение 'информатика'.

При присваивании переменной строкового типа значения строки символов, происходит усечение строки символов справа, если ее длина превышает длину, указанную в определении типа строковой переменной.

2.2. Операции сравнения (отношения)

Это бинарные операции со знаками =|<>|<|>|<=|>= . Операнды любого строкового типа. Результат типа BOOLEAN. Сравнение производится посимвольно в соответствии с кодами символов, начиная с левого края строк, первое несовпадение определяет результат операции (отсутствие символа - код 0). 

Для определения результата операции сравнения можно использовать два следующих правила:

1) две строки равны, если их динамические длины равны и попарно совпадают все соответствующие символы обеих строк;

2) первая строка считается меньше второй, если после первых попарно совпадающих символов код символа первой строки меньше кода символа второй строки или в первой строке отсутствует символ.

Например, сравнение 'FRE'>'FR' дает результат TRUE, т.к. первые две литеры в операндах совпадают, а третья литера во втором операнде отсутствует, что и определяет результат (второй операнд меньше первого); сравнение 'ALFAVIT'>'GAMMA' даст результат FALSE, так как код литеры 'A' меньше кода литеры 'G'.

2.3. Операция определения адреса

Унарная операция со знаком @. Операнд - переменная строкового типа, результат - указатель на участок оперативной памяти, в котором располагается переменная (адрес нулевого байта строки).

2.4. Подпрограммы

2.4.1. Функции

а) определение текущей (динамической длины) строки:

LENGTH(<исходная строка>:STRING):INTEGER;

б) выделение из строки подстроки (получение частичной копии):

COPY(<исходная строка>:STRING;
     <номер начальной позиции>,
     <количество копируемых символов>:INTEGER)

:STRING;

в) определение позиции первого вхождения подстроки в строку (если подстрока не входит в строку, то результат 0):

POS(<подстрока>,<строка>:STRING):BYTE;

г) конкатенация (слияние) группы строк:

CONCAT(<первая строка>,

       <вторая строка>,

        ...

       <последняя строка>:STRING)

:STRING;

Таблица 21.1. Примеры использования функций:

	Операция
	Результат

	LENGTH('Информатика')
	11

	COPY('Информатика',3,5)

COPY('Информатика',3,100)

COPY('Информатика',30,5)
	'форма'

'форматика'

''

	POS('форма','Информатика')

POS('реформа','Информатика')

POS('ма','мама')
	3

0

1

	CONCAT('при','м','ус','ы')
	'примусы'


2.4.2. Процедуры  редактирования строк

а) удаление последовательности символов строке:

DELETE(VAR <обрабатывааемая строка>:STRING;

       <номер начальной позиции>:INTEGER;

       <количество удаляемых символов>:INTEGER);

Например, при выполнении операторов

a:='ИНФОРМАТИКА';

DELETE(a,3,6)
переменная а получит значение 'ИНИКА'.

При выполнении операторов

a:='ИНФОРМАТИКА';

DELETE(a,3,60)

переменная а получит значение 'ИН'.

б) вставка подстроки в строку:

INSERT(<вставляемая строка>:STRING;

       VAR <обрабатывааемая строка>:STRING;

       <номер начальной позиции>:INTEGER).

Например, при выполнении операторов

a:='ИНИКА';

INSERT('формат',a,3)

переменная а получит значение 'ИНформатИКА'.

Процедура работает, как конкатенация, если номер начальной позиции больше, чем длина обрабатываемой строки (подстрока "приклеивается" справа к строке);

2.4.3. Процедуры преобразования строк

а) преобразование численного значения в строку символов:

STR(<число>:<любой целый или вещественный тип>;

    VAR <результирующая строка>:STRING ) .

При обращении к процедуре первый фактический параметр может через двоеточие сопровождать формат - ширина поля для числа и число знаков после десятичной точки (как при обращении к процедуре вывода значений);

б) преобразование строки символов в число:

VAL(<исходная строка>:STRING;

    VAR <результирующее число>:<любой целый или вещественный тип>;

    VAR <код ошибки>:INTEGER).

Процедура преобразует числовое значение, записанное в исходной строке, в значение числовой переменной. Если преобразование возможно, то формируется нулевое значение переменной-кода ошибки, в противном случае эта переменная содержит номер символа в исходной строке, который не может входить в запись численной величины.

21.4. Пример обработки строк

Постановка задачи

Строка символов состоит из слов, в которые входят только русские буквы. Оканчивается строка точкой. Одно слово от другого отделяется одним или несколькими пробелами. Оставить в строке только те слова, в которых  чередуются гласные и согласные буквы, и  этими слова симметричны относительно своего центра.

Например, если исходная строка 

МИМО ЦИРКА МИМ ИМАМ ОГОГО ОСЕЛ ЛЕТЕЛ.КОК

то результирующая строка 

МИМ ОГОГО ЛЕТЕЛ

Структура программы 

Предлагается задачу разбить на семь подзадач. Выделенные подзадачи приведены на структурной диаграмме программы (рис. 21.3).

Рис. 21.3. – Структурная диаграмма программы

Разработка подпрограммы 1

Спецификация

1. Назначение: замена в строке каждого вхождения заданного фрагмента на новое значение (в нашей программе будет использоваться для подавления повторяющихся пробелов, то есть для того, чтобы между словами оставить только один пробел)

2. Имя: redfrag

3. Вид: процедура

4. Перечень параметров:

Таблица 21.2. Перечень параметров

	Статус
	Назначение
	Имя
	Тип
	Вид

	Вход/выход
	Обрабатываемая строка
	s
	string
	параметр-переменная

	Вход
	Исходный фрагмент
	s1
	string
	параметр-значение

	Вход
	Новый фрагмент
	s2
	string
	параметр-значение


5. Заголовок п/п:  procedure redfrag(var s:string; s1,s2:string);
Метод решения

Словесно метод решения сформулируем так:

Пока исходный фрагмент s1 входит в строку s повторять

а) зафиксировать местоположение первого вхождения исходного фрагмента s1 в строку s;

б) удалить из строки s, начиная с зафиксированной позиции, столько символов, какова длина  исходного фрагмента s1;

в) вставить фрагмент s2 в строку s , начиная с зафиксированной позиции

Этот метод можно записать следующим образом:

Пока pos(s1,s)<>0 :

Информационная модель

Таблица 21.3. Информационная модель

	Назначение
	Имя
	Тип

	Номер позиции в строке, начиная с которого фрагмент входит в строку
	k
	integer


Программная модель

procedure redfrag(var s:string;s1,s2:string);

var k:integer;

begin

  while pos(s1,s)<>0 do

    begin

      k:=pos(s1,s);

      delete(s,k,length(s1));

      insert(s2,s,k)

    end

end;

Разработка подпрограммы 2

Спецификация

1. Назначение: ввод исходной строки с клавиатуры и контроль за тем, что она состоит из русских букв и  пробелов, разделяющих слова. Оканчивается строка точкой, после которой символы не обрабатываются. При успешном вводе строки формируется значение истина, при неуспешном – ложь. Строка подготавливается для дальнейшей обработки – удаляются завершающая точка и все символы, следующие за ней, в конец вставляется пробел (для того, чтобы каждое слово оканчивалось пробелом), подавляются повторяющиеся пробелы (для того, чтобы слово оканчивалось только одним пробелом).

2. Имя: control

3. Вид: функция

4. Перечень параметров 

Таблица 21.4. Перечень параметров

	Статус
	Назначение
	Имя
	Тип
	Вид

	Выход
	Заданная вводом строка
	s
	string
	параметр-переменная

	Вход
	Множество символов, из которых может состоят вводимая строка (исключая точку и символы, расположенные после нее)
	simb
	tsetchar
	параметр-константа

	Возвращаемый результат
	Признак правильности ввода строки (истина –успешный ввод, в противном случае –ложь)
	control
	boolean
	-


Примечание: type tsetchar=set of char;
5. Заголовок п/п: 
function control(var s:string; const simb: tsetchar) :Boolean ;

Метод решения

1. Ввод с клавиатуры строки s;

2. Если точка отсутствует в строке или располагается в ее начале, то формируются возвращаемый результат со значением ложь и пустое значение выходной строки и осуществляется выход из подпрограммы

3. Для дальнейшей обработки исходная строка усекается на месте завершающей точки и в конец добавляется пробел, который является признаком окончания слова

s:=copy(s,1,pos('.',s)-1) + ' '
;
4. В исходной строке удаляются повторяющиеся пробелы – между словами остается по одному пробелу. Это осуществляется с помощью подпрограммы замены в заданной строке фрагмента из двух подряд идущих пробела на один пробел

redfrag(s,'  ',' ')
;
5. Перебираются все символы и проверяется их принадлежность множеству возможных символов simb. Если очередной символ не принадлежит этому множеству то формируются возвращаемый результат со значением ложь и пустое значение выходной строки и осуществляется выход из подпрограммы


[image: image6.wmf]ê

ê

ê

ë

é

=

=

Þ

Ì

=

"

exit

s

false

control

simb

i

s

если

s

length

i

'

'

:

:

]

[

:

)

(

..

1

:


; 

6. Успешное завершение ввода строки

control:=true
Информационная модель

Таблица 21.5 . Информационная модель

	Назначение
	Имя
	Тип

	Индекс символа в строке
	i
	integer


Программная модель

function control(var s:string; const simb:tsetchar):boolean;

var i:integer;

begin
{ввод строки с клавиатуры}

  writeln('Исходная строка?');

  readln(s);

{проверка наличия и местоположения точки; если точки нет или она стоит в начале строки, то формируем ошибку при вводе и выходим из подпрограммы}

  if pos('.',s)<=1 then

    begin

      control:=false;

      s:='';

      exit

    end;

{формирование обрабатываемой части строки (до точки) и добавление пробела в конец строки}

  s:=copy(s,1,pos('.',s)-1)+' ';

{подавление повторяющихся пробелов – остается по одному пробелу}

  redfrag(s,'  ',' ');

{перебор всех символов строки и проверка их принадлежности множеству допустимых символов}

  for i:=1 to length(s) do

    {если очередной символ не входит в множество возможных}

    if not (s[i] in simb) then

       {ошибочный ввод исходной строки}

       begin

         control:=false;

         s:='';

         exit

       end;

{успешное завершение ввода строки}

  control:=true

end;
Разработка подпрограммы 3

Спецификация

1. Назначение: вычисление количества слов в строке (каждое слово в строке оканчивается одним пробелом)

2. Имя: kolslov
3. Вид: функция

4. Перечень параметров

Таблица 21.6. Перечень параметров

	Статус
	Назначение
	Имя
	Тип
	Вид

	Вxoд
	Исходная строка слов 
	s
	string
	параметр-значение

	Возвращаемый результат
	Количество слов в строке
	kolslov
	integer
	-


5. Заголовок п/п:   function kolslov(s:string):integer;
Метод решения

1. Обнуление счетчика слов

sum:=0
;

2. Перебор всех символов строки и при обнаружении очередного пробела увеличение счетчика слов на 1, так как пробел является признаком завершения слова

3. Формирование возвращаемого результата функции

kolslov:=sum
Информационная модель

Таблица 21.7. Информационная модель

	Назначение
	Имя
	Тип

	Индекс символа в строке
	i
	integer

	Счетчик пробелов
	sum
	integer


Программная модель
function kolslov(s:string):integer;

var i,sum:integer;

begin

  sum:=0;

  for i:=1 to length(s)do

    if s[i]=' ' then sum:=sum+1;

  kolslov:=sum

end;

Разработка подпрограммы 4

Спецификация

1. Назначение: выделение из строки слова, заданного его порядковым номером (считается, что все слова пронумерованы подряд, начиная с 1). Кроме этого, формируются номер позиции в строке, начиная с которой в строке располагается искомое слово, и длина этого слова. Если слово, с указанным номером в строке отсутствует, то возвращаемый результат – пустая строка, а номер позиции и длина слова 0.

2. Имя: slovo
3. Вид: функция

4.Перечень параметров

Таблица 21.8. Перечень параметров

	Статус
	Назначение
	Имя
	Тип
	Вид

	Вxoд
	Исходная строка слов 
	s
	string
	параметр-значение

	Вход
	Номер искомого слова
	k
	integer
	параметр-значение

	Выход
	Номер позиции в строке, с которой начинается слово
	wordpos
	integer
	параметр-переменная

	Выход
	Длина слова (количество символов в слове)
	wordsize
	integer
	параметр-переменная

	Возвращаемый результат
	Искомое слово
	slovo
	string
	-


5. Заголовок п/п:

function slovo(s:string; k:integer;var wordpos,wordsize:integer):string;

Метод решения

1. Обнуление счетчика количества обнаруженных слов

kol:=0
;

2. Установка индекса просматриваемого символа строки в 1

i:=1
;

3. Обнуление счетчика подсчета длины найденного слова

wordsize:=0
;

4. Установка на начало первого слова

wordpos:=1
;

5. Пока не нашли нужное по номеру слово и не вышли за пределы заданной строки повторять

а) если очередной символ пробел, то количество найденных слов увеличить на 1 и после этого проверить – если количество обнаруженных слов еще не достигло номера искомого слова, то обнулить размер найденного слова и определить начало нового слова как позицию пробела плюс 1. 

В противном случае (если очередной символ не пробел), то увеличить длину текущего слова на 1

б) перейти к следующему символу строки

i:=i+1
;

6. Если по выходу из цикла количество обнаруженных слов равняется номеру необходимого слова, то результату функции присвоим выделяемое из строки слово, в противном случае (слово с заданным номером не существует) результату функции присваиваем пустую строку и обнуляем выходные параметры (номер позиции, с которой начинается слово, и количество символов в слове)

Информационная модель

Таблица 21.9. Информационная модель

	Назначение
	Имя
	Тип

	Индекс символа в строке
	i
	integer

	Счетчик слов
	kol
	integer


Программная модель
function slovo(s:string;k:integer;var wordpos,wordsize:integer):string;

var kol,i:integer;

begin

  kol:=0;

  i:=1;

  wordsize:=0;

  wordpos:=1;

  while(kol<>k)and(i<=length(s)) do

     begin

       if s[i]=' ' then

         begin

           kol:=kol+1;

           if kol<>k then

              begin

                wordsize:=0;

                wordpos:=i+1

              end

          end

       else wordsize:=wordsize+1;

       i:=i+1

     end;

  if kol=k then

       slovo:=copy(s,wordpos,wordsize)

  else

     begin

       slovo:='';

       wordpos:=0;

       wordsize:=0

     end

end;

Разработка подпрограммы 5

Спецификация

1. Назначение: проверка симметричности слова относительно его центра. Результат истина, если слово симметрично, и ложь в противном случае.

2. Имя: simmetr
3. Вид: функция

4. Перечень параметров

Таблица 21.10. Перечень параметров

	Статус
	Назначение
	Имя
	Тип
	Вид

	Вxoд
	Исходное слово 
	s
	string
	параметр-значение

	Возвращаемый результат
	Признак симметричности слова (истина – слово симметрично, ложь –несимметрично) 
	simmetr
	boolean
	-


5. Заголовок п/п: function simmetr(s:string):boolean;

Метод решения

1. Результату функции присвоить истина, так как пока не найдены несимметричные буквы

simmetr:=true
;

2. Перебирать номера букв от единицы до половины строки символов. Если очередная буква левой половины не совпадает с симметрично расположенной буквой в правой половине слова, то результату функции присвоить ложь

Информационная модель

Таблица 21.11. Информационная модель

	Назначение
	Имя
	Тип

	Индекс символа в строке
	i
	integer


Программная модель

function simmetr(s:string):boolean;

var i:integer;

begin

  simmetr:=true;

  for i:=1 to length(s) div 2 do

    if s[i]<>s[length(s)-i+1] then simmetr:=false

end;

Разработка подпрограммы 6

Спецификация

1. Назначение: проверка чередования гласных и согласных в строке. Результат истина, если все буквы слова чередуются, и ложь в противном случае.

2. Имя: chered

3. Вид: функция

4. Перечень параметров

Таблица 21.12. Перечень параметров

	Статус
	Назначение
	Имя
	Тип
	Вид

	Вxoд
	Исходное слово 
	s
	string
	параметр-значение

	Возвращаемый результат
	Признак чередования букв(истина – гласные и согласные чередуются, ложь –нет) 
	chered
	boolean
	-


5. Заголовок п/п:  function chered(s:string):boolean;

Метод решения

1. Результату функции присвоить истина, так как пока не найдены нечередующиеся буквы

chered:=true
;

2. Определение ожидающейся буквы. Если первый символ принадлежит множеству гласных букв (множество mn1), то ожидаемая буква – согласная (переменной next присваивается значение истина), в противном случае ожидаемая буква – гласная (переменной next присваивается значение ложь)

3. Перебор всех остальных букв, начиная со 2 по конец слова. Если очередная буква принадлежит множеству согласных букв (mn2) и ожидалась согласная, то следующая ожидаемая – гласная буква (next присваивается ложь). В противном случае, если очередная буква принадлежит множеству гласных букв (mn1) и ожидалась гласная, то следующая ожидаемая – согласная буква (next присваивается истина); в противном случае чередование нарушено (результату функции присваивается ложь)

Информационная модель

Таблица 21.13. Информационная модель

	Назначение
	Имя
	Тип

	Индекс символа в строке
	i
	integer

	Признак ожидаемой буквы (истина – ожидается согласная буква, ложь – гласная)
	next
	boolean


Программная модель

function chered(s:string):boolean;

const mn1:tsetchar=['А','И','Е','Ё','О','У','Э','Ы','Ю','Я',

                    'а','и','е','ё','о','у','э','ы','ю','я'];

      mn2:tsetchar=['Б','В','Г','Д','Ж','З','Й','К','Л','М','Н',

                    'П','Р','С','Т','Ф','Х','Ц','Ч','Ш','Щ',

                    'б','в','г','д','ж','з','й','к','л','м','н',

                    'п','р','с','т','ф','х','ц','ч','ш','щ'];

var i:integer;

    next:boolean;

begin

  chered:=true;

  if s[1] in mn1 then next:=true else next:=false;

  for i:=2 to length(s) do

    if (s[i] in mn2) and next then

      next:=false

    else

      if (s[i] in mn1) and not next then

        next:=true

      else chered:=false

end;

Разработка подпрограммы 7

Спецификация

1. Назначение: удаление из заданной строки слов, которые не состоят из чередующихся гласных и согласных букв, и не являются симметричными относительно своего центра.

2. Имя: udalen

3. Вид: процедура

4. Перечень параметров

Таблица 21.14. Перечень параметров

	Статус
	Назначение
	Имя
	Тип
	Вид

	Вxoд/выход
	Обрабатываемая строка слов
	s
	string
	параметр-переменная


5. Заголовок п/п:  procedure udalen(var s:string);
Метод решения

1. Для всех слов строки проверяем, если очередное k-ое слово не является последовательностью чередующихся гласных и согласных или не является симметричным относительно центра слова, то удаляем его (при этом величина k не изменяется). В противном случае увеличиваем номер слова k на 1. Так как в процессе перебора количество слов может изменяться, то использовать цикл с параметром нельзя. Поэтому метод решения следующий

Информационная модель

Таблица 21.16. Информационная модель

	Назначение
	Имя
	Тип

	Номер слова в строке
	k
	integer

	Позиция начала слова в строке
	beg
	integer

	Количество символов в слове
	kol
	integer


Программная модель

procedure udalen(var s:string);

var  k,beg,kol:integer;

begin

  k:=1;

  while k<=kolslov(s) do

       if not chered(slovo(s,k,beg,kol)) or

          not simmetr(slovo(s,k,beg,kol))then

          delete(s,beg,kol+1)

       else k:=k+1;

end;

Разработка программы

Метод решения

1. Если при вводе строки ее символы не русские буквы и не пробелы, то строка задана неверно – сообщаем об этом пользователю и прекращаем выполнение программы

2. удаление из заданной строки слов, которые не состоят из чередующихся гласных и согласных букв, и не являются симметричными относительно своего центра

udalen(s)
;

3. Выводим результирующую строку. Если она пуста, то сообщаем об этом.

Информационная модель

Таблица 21.17. Информационная модель

	Статус
	Назначение
	Имя
	Тип

	Вход/выход
	Строка слов
	s
	string


Набор тестов
	№
	Исходная строка
	Ожидаемый результат

	1
	'МИМО ЦИРКА МИМ ИМАМ ОГОГО ОСЕЛ ЛЕТЕЛ.КОК'
	'МИМ ОГОГО ЛЕТЕЛ'

	2
	'ПРООРП ПРИВЕТ.'
	Пустая строка


Программная модель (текст программы на языке Паскаль)

program primer;

type

  tsetchar=set of char;

{здесь располагаются тексты подпрограмм}

var s:string;

begin

  if not control(s,['А'..'Щ','Ы','Э'..'Я','Ё','а','б','в','г','д','е','ё','ж','з','и','й','к','л','м','н',о,п,р,с,т,у,ф,х,ц,ч,ш,щ,ы,э,ю,'я',' ']) then

    begin

      writeln('ОШИБКА при вводе строки!!!');

      halt

    end;

  udalen(s);

  writeln('РЕЗУЛЬТАТ:');

  if length(s)=0 then 

    writeln('Нет слов, удовлетворяющих одновременно условиям чередования и симметричности')

  else 

    writeln('Результирующая строка: ',s)

end.

